Equations and Stock-Flow Diagrams Omitted from the Main Text Due to File Size Limit



Stock-Flow Diagram of the Closed-Loop Supply Chain



Stock-Flow Diagram of the *Legislation*

Models' Equations

Levels

init Collected_Products = 3
flow Collected_Products=-dt*Products_Accepted_for_Recycling+dt*Collection_Rate -dt*Products_Rejected_for_Recycling
doc Collected_Products = Products arriving at the collection centers.
unit Collected_Products = items

init Demand_Backlog = 0
flow Demand_Backlog = -dt*Demand_Backlog_Reduction_Rate +dt*Demand
doc Demand_Backlog = Unsatisfied demand served when Distributor's_Inventory is available.
unit Demand_Backlog = items

init Disposed_Products = 0
flow Disposed_Products=+dt*Products_Rejected_for_Recycling +dt*Controllable_Disposal
doc Disposed_Products = Accumulates the controllably disposed products.
unit Disposed_Products = items

init Distributors_Inventory = 0
flow Distributors_Inventory = -dt*Sales +dt*Shipments_to_Distributor
doc Distributors_Inventory = Distributor's on-hand inventory.
unit Distributors_Inventory = items

dim Legislation = (L=1..4)
init Legislation = [0.8,0.87,0.87,0]
flow Legislation = +dt*Change_in_Legislation
doc Legislation = The initial values of the environmental legislation in 2003.

1: Collection Percentage
2: Recycling Percentage
3: Recyclability
4: Recycled Content

unit Legislation = (dimensionless)

init Natural_Resources = Initial_Resources
flow Natural_Resources = -dt*ARRSUM(Procurement_Rate)
doc Natural_Resources = Inventory of natural resources.
unit Natural_Resources = items

init Orders_Backlog = 0
flow Orders_Backlog=-dt*Orders_Backlog_Reduction_Rate+dt*Distributors_Orders
doc Orders_Backlog = Unsatisfied Distributor's_Orders served when Serviceable_Inventory is available.
unit Orders_Backlog = items

init Production_Backlog = 0
flow Production_Backlog=-dt*Production_Backlog_Reduction_Rate +dt*Producers_Orders
doc Production_Backlog = Unsatisfied Producer's_Orders served when raw materials are available.
unit Production_Backlog = items

dim Raw_Materials_Inventory = (R=1..2)
init Raw_Materials_Inventory = [0,100]
flow Raw_Materials_Inventory = +dt*Recycling_Rate +dt*Procurement_Rate -dt*Usage_Rate
doc Raw_Materials_Inventory = Inventory of producer's natural resources and recycled materials.
1: Non-renewable materials
2: Recycled materials
unit Raw_Materials_Inventory = items

init Recyclable_Products = 0
flow Recyclable_Products = -dt*ARRSUM(Recycling_Rate) +dt*Products_Accepted_for_Recycling -dt*Controllable_Disposal
doc Recyclable_Products = Inventory of Used_Products that passed inspection and can be recycled.
unit Recyclable_Products = items

init Serviceable_Inventory = 0
flow Serviceable_Inventory = +dt*Production_Rate-dt*Shipments_to_Distributor
doc Serviceable_Inventory = On-hand inventory of new products.
unit Serviceable_Inventory = items

init Sum_Disposal = 0
flow Sum_Disposal = +dt*Products_Rejected_for_Recycling

+dt*Uncontrollable_Disposal+dt*Controllable_Disposal
doc Sum_Disposal = Accumulates the Products_Rejected_for_Recycling, Controllable_Disposal and Uncontrollable_Disposal.
unit Sum_Disposal = items

init Uncontrollably_Disposed_Products = 0
flow Uncontrollably_Disposed_Products = +dt*Uncontrollable_Disposal
doc Uncontrollably_Disposed_Products = Accumulates the uncontrollably disposed products.
unit Uncontrollably_Disposed_Products = items

Rates
dim Change_in_Legislation = (L=1..4)
aux Change_in_Legislation=PULSE(MAX(Desired_Collection-Legislation(1),0), Leg_Time,Leg_Time)*Switch1|L=1;PULSE(MAX(Desired_Recycling-Legislation(2),0),Leg_Time,Leg_Time)*Switch1|L=2;PULSE(MAX (Desired_Recycling-Legislation(3),0),Leg_Time,Leg_Time)*Switch1|L=3; PULSE(MAX(Desired_Content-Legislation(4),0),Leg_Time,Leg_Time)* Switch1 | L=4
doc Change_in_Legislation = Change in the current Legislation.
1: Collection Percentage
2: Recycling Percentage
3: Recyclability
4: Recycled Content
unit Change_in_Legislation = 1/week

aux Collection_Rate=MIN(Collection_Capacity,MAX(Used_Products *Collection_Percentage,Used_Products*Minimum_Collection_Percentage))
doc Collection_Rate = Flow of Used_Products to collection facilities.
unit Collection_Rate = items/week

aux Controllable_Disposal=MAX(Recyclable_Products-Recycling_Rate(2) *TIMESTEP,0)/Recyclable_Stock_Keeping_Time
doc Controllable_Disposal = Flow of surplus stock of Recyclable_Products to prevent the costly accumulation if Recycling_Capacity is not enough.
unit Controllable_Disposal = items/week

aux Demand = ((1+0.000791)^TIME)*NORMAL(13.85,3.58,0.01)
doc Demand = Demand of the specific firm.
unit Demand = items/week

aux Demand_Backlog_Reduction_Rate = Sales
doc Demand_Backlog_Reduction_Rate = The reduction of the Demand_Backlog due to sales.
unit Demand_Backlog_Reduction_Rate = items/week

aux Distributors_Orders = Expected_Demand+DI_Discrepancy/DI_Adj_Time
doc Distributors_Orders = Orders placed from the distributor to producer.

unit Distributors_Orders = items/week

aux Orders_Backlog_Reduction_Rate = Shipments_to_Distributor
doc Orders_Backlog_Reduction_Rate = The reduction of the Orders_Backlog due to Shipments_to_Distributor.
unit Orders_Backlog_Reduction_Rate = items/week

dim Procurement_Rate = (R=1..2)
aux Procurement_Rate=IF(Raw_Materials_Inventory(2)/TIMESTEP <Recycled_Content*Expected_Raw_Materials, MAX(MIN(Natural_Resources /Procurement_Time, Expected_Raw_Materials-Raw_Materials_Inventory(2) /TIMESTEP),0),MAX(MIN(Natural_Resources/Procurement_Time,((1-Recycled_Content)*Expected_Raw_Materials)-((Raw_Materials_Inventory(2)/ TIMESTEP)-Recycled_Content*Expected_Raw_Materials)),0))|R=1;0|R=2
doc Procurement_Rate = Flow of Natural_Resources to the producer.
1: Natural resources
2: Recycled materials
unit Procurement_Rate = items/week

aux Producers_Orders=Expected_Distributors_Orders +SI_Discrepancy/SI_Adj_Time
doc Producers_Orders = Orders placed from the producer to raw materials' warehouse.
unit Producers_Orders = items/week

aux Production_Backlog_Reduction_Rate = ARRSUM(Usage_Rate)
doc Production_Backlog_Reduction_Rate = The reduction of the Production_Backlog due to Production_Rate.
unit Production_Backlog_Reduction_Rate = items/week

aux Production_Rate = ARRSUM(Usage_Rate)
doc Production_Rate = The firm's production rate (it includes both natural resources and recycled materials as raw materials).
unit Production_Rate = items/week

aux Products_Accepted_for_Recycling=MAX(MIN(Collected_Products *Recycling_Percentage,Collected_Products*Recyclability),Collected_Products *Minimum_Recycling_Percentage)/Inspection_Time
doc Products_Accepted_for_Recycling = Flow of Used_Products that have passed inspection and can be recycled.
unit Products_Accepted_for_Recycling = items/week

aux Products_Rejected_for_Recycling=MAX(Collected_Products-Products_Accepted_for_Recycling*TIMESTEP,0)/Inspection_Time
doc Products_Rejected_for_Recycling = Flow of Used_Products that have not passed inspection and should be disposed.
unit Products_Rejected_for_Recycling = items/week

dim Recycling_Rate = (R=1..2)
aux Recycling_Rate = 0 | R=1 ; MIN(Recyclable_Products/TIMESTEP, Recycling_Capacity) | R=2
doc Recycling_Rate = Flow of Recyclable_Products through the recycling facilities.
1: Natural resources
2: Recycled materials
unit Recycling_Rate = items/week

aux Sales = MIN(Demand_Backlog,Distributors_Inventory)/Delivery_Time
doc Sales = Sales according to demand and Distributor's_Inventory.
unit Sales = items/week

aux Shipments_to_Distributor=MIN(Serviceable_Inventory,Orders_Backlog) /Shipment_Time
doc Shipments_to_Distributor = Shipments_to_Distributor according to distributor's orders and Serviceable_Inventory.
unit Shipments_to_Distributor = items/week

aux Uncontrollable_Disposal = MAX(Used_Products-Collection_Rate,0)
doc Uncontrollable_Disposal = Flow of Used_Products to disposal due to limited Collection_Capacity.
unit Uncontrollable_Disposal = items/week

dim Usage_Rate = (R=1..2)
aux Usage_Rate=MIN(Raw_Materials_Inventory(1)/TIMESTEP, Accomplishable_Production_Rate-MIN(Accomplishable_Production_Rate, Raw_Materials_Inventory(2)/TIMESTEP))|R=1;MIN(Accomplishable_ Production_Rate,Raw_Materials_Inventory(2)/TIMESTEP) | R=2
doc Usage_Rate = The usage rate of raw materials.
1: Natural resources
2: Recycled materials
unit Usage_Rate = items/week

Auxiliaries

aux Accomplishable_Production_Rate=MAX(MIN(Production_Capacity, Production_Backlog/TIMESTEP),0)
doc Accomplishable_Production_Rate = Ancillary variable for the formulation of Production_Rate.
unit Accomplishable_Production_Rate = items/week

aux Availability_of_Resources = Natural_Resources DIVZ0 Initial_Resources
doc Availability_of_Resources = Fraction of Natural_Resources to Initial_Resources.
unit Availability_of_Resources = (dimensionless)

aux AvrLandfill = DELAYINF(Landfill_Availability,a_L,1,Landfill_Availability)
doc AvrLandfill = Forecast of Landfill_Availability obtained using exponential smoothing.

```
unit    AvrLandfill = (dimensionless)

aux     AvrResAv=DELAYINF(Availability_of_Resources,a_AR,1,
        Availability_of_Resources)
doc     AvrResAv = Forecast of Availability_of_Resources obtained using exponential
        smoothing.
unit    AvrResAv = (dimensionless)

aux     Collection_Percentage = Legislation(1)
doc     Collection_Percentage = Take-back obligations.
unit    Collection_Percentage = (dimensionless)

aux     Desired_Collection = GRAPH(AvrLandfill,0,1,[1,0"Min:0;Max:1"]) *Tactics(1)
        +GRAPH(AvrLandfill,0,0.1,[1,0.6,0.34,0.22,0.16,0.12,0.1,0.08,0.06,0.03,0
        "Min:0;Max:1"])*Tactics(2)  +GRAPH(AvrLandfill,0,0.1,[1,0.99,0.97,0.93,0.89,
        0.82,0.74,0.64,0.51,0.33,0"Min:0;Max:1"])*Tactics(3)  +GRAPH(AvrLandfill,0,
        0.1,[1,0.74,0.63,0.55,0.52,0.5,0.48,0.45,0.37,0.26,0"Min:0;Max:1"])*Tactics(4)
doc     Desired_Collection = Desired change in current Collection Percentage.
unit    Desired_Collection = (dimensionless)

aux     Desired_Content = GRAPH(AvrResAv,0,1,[1,0"Min:0;Max:1"])*Tactics(1)+
        GRAPH(AvrResAv,0,0.1,[1,0.6,0.34,0.22,0.16,0.12,0.1,0.08,0.06,0.03,0
        "Min:0;Max:1"])*Tactics(2)+GRAPH(AvrResAv,0,0.1,[1,0.99,0.97,0.93,0.89,
        0.82,0.74,0.64,0.51,0.33,0"Min:0;Max:1"])*Tactics(3)    +GRAPH(AvrResAv,0,
        0.1,[1,0.74,0.63,0.55,0.52,0.5,0.48,0.45,0.37,0.26,0"Min:0;Max:1"])*Tactics(4)
doc     Desired_Content = Desired change in current Recycled Content.
unit    Desired_Content = (dimensionless)

aux     Desired_DI = Expected_Demand*DI_Cover_Time
doc     Desired_DI = Desired distributor's inventory.
unit    Desired_DI = items

aux     Desired_Recycling = GRAPH(Rec_Ratio,0,1,[1,0"Min:0;Max:1"])*Tactics(1)+
        GRAPH(Rec_Ratio,0,0.1,[1,0.6,0.34,0.22,0.16,0.12,0.1,0.08,0.06,0.03,0
        "Min:0;Max:1"])*Tactics(2)    +GRAPH(Rec_Ratio,0,0.1,[1,0.99,0.97,0.93,0.89,
        0.82,0.74,0.64,0.51,0.33,0"Min:0;Max:1"])*Tactics(3)    +GRAPH(Rec_Ratio,0,
        0.1,[1,0.74,0.63,0.55,0.52,0.5,0.48,0.45,0.37,0.26,0"Min:0;Max:1"])*Tactics(4)
doc     Desired_Recycling = Desired change in current recycling activities.
unit    Desired_Recycling = (dimensionless)

aux     Desired_RM = Expected_Producers_Orders*RM_Cover_Time
doc     Desired_RM = Desired natural materials' inventory.
unit    Desired_RM = items

aux     Desired_SI = Expected_Distributors_Orders*SI_Cover_Time
doc     Desired_SI = Desired serviceable inventory.
unit    Desired_SI = items
```

aux DI_Discrepancy = MAX(Desired_DI-Distributors_Inventory,0)
doc DI_Discrepancy = Discrepancy between the desired distributor's inventory and the actual distributor's inventory.
unit DI_Discrepancy = items

aux Expected_Demand = DELAYINF(Demand,a_D,1,Demand)
doc Expected_Demand = Exponential smoothing of Demand.
unit Expected_Demand = items/week

aux Expected_Distributors_Orders=DELAYINF(Distributors_Orders,a_DI,1, Distributors_Orders)
doc Expected_Distributors_Orders = Exponential smoothing of Distributor's orders.
unit Expected_Distributors_Orders = items/week

aux Expected_Producers_Orders=DELAYINF(Producers_Orders,a_SI,1, Producers_Orders)
doc Expected_Producers_Orders = Exponential smoothing of Producer's_orders.
unit Expected_Producers_Orders = items/week

aux Expected_Raw_Materials=Expected_Producers_Orders-Expected_Recycling_Rate+RM_Discrepancy/RM_Adj_Time
doc Expected_Raw_Materials = Ancillary variable for the formulation of Procurement_Rate.
unit Expected_Raw_Materials = items/week

aux Expected_Recycling_Rate=DELAYINF(Recycling_Rate(2),a_RR,1, Recycling_Rate(2))
doc Expected_Recycling_Rate = Exponential smoothing of Recycling_rate.
unit Expected_Recycling_Rate = items/week

aux Landfill_Availability = MAX((Initial_Landfill_Availability-Disposed_Products -Uncontrollably_Disposed_Products),0) DIVZ0 Initial_Landfill_Availability
doc Landfill_Availability = Reflects how much the available landfills have shrunk concerning the Initial_Landfill_Availability.
unit Landfill_Availability = (dimensionless)

aux Limit_of_Recyclability = Legislation(3)
doc Limit_of_Recyclability = Legislative recyclability.
unit Limit_of_Recyclability = (dimensionless)

aux Limit_of_Recycled_Content = Legislation(4)
doc Limit_of_Recycled_Content = Legislative recycled content.
unit Limit_of_Recycled_Content = (dimensionless)

aux Rec_Ratio = AvrResAv*AvrLandfill
doc Rec_Ratio = Product of AvrLandfill and AvrResAv.
unit Rec_Ratio = (dimensionless)

aux Recyclability = MAX(DELAYMTR(Limit_of_Recyclability, Redesign_Time, 3, Minimum_Recyclability), Minimum_Recyclability)
doc Recyclability = If a product can be recycled.
unit Recyclability = (dimensionless)

aux Recycled_Content=MAX(Limit_of_Recycled_Content, Minimum_Limit_of_Recycled_Content)
doc Recycled_Content = Percentage of recycled materials to be used in producing goods.
unit Recycled_Content = (dimensionless)

aux Recycling_Percentage = Legislation(2)
doc Recycling_Percentage = Legislative recycling percentage.
unit Recycling_Percentage = (dimensionless)

aux Residence_Time = NORMAL(16.444,4.1995,0.29)
doc Residence_Time = Distribution of residence time of refrigerators.
unit Residence_Time = year

aux RM_Discrepancy=MAX(Desired_RM-(Raw_Materials_Inventory(1)+Raw_Materials_Inventory(2)),0)
doc RM_Discrepancy = Discrepancy between the desired raw materials' inventory and the actual raw materials' inventory.
unit RM_Discrepancy = items

aux SI_Discrepancy = MAX(Desired_SI-Serviceable_Inventory,0)
doc SI_Discrepancy = Discrepancy between the desired serviceable inventory and the actual serviceable inventory.
unit SI_Discrepancy = items

aux Used_Products = DELAYMTR(Sales,Residence_Time*50,20,10)
doc Used_Products = Products in use.
unit Used_Products = items/week

Constants
const a_AR = 25
doc a_AR = Smoothing factor used for the estimation of AvrResAv.
unit a_AR = week

const a_D = 2
doc a_D = Smoothing factor used for the estimation of Demand.
unit a_D = week

const a_DI = 2
doc a_DI = Smoothing factor used for the estimation of Distributor's_Inventory.
unit a_DI = week

const a_L = 25

doc a_L = Smoothing factor used for the estimation of AvrLandfill.
unit a_L = week

const a_RR = 25
doc a_RR = Smoothing factor used for the estimation of Raw_Materials_Inventory.
unit a_RR = week

const a_SI = 2
doc a_SI = Smoothing factor used for the estimation of Serviceable_Inventory.
unit a_SI = week

const Collection_Capacity = 1000
doc Collection_Capacity = Capacity of the collection facilities.
unit Collection_Capacity = items/week

const Delivery_Time = 2
doc Delivery_Time = Time needed to transfer products from distributor to end users.
unit Delivery_Time = week

const DI_Adj_Time = 1
unit DI_Adj_Time = week

const DI_Cover_Time = 2
unit DI_Cover_Time = week

const Initial_Landfill_Availability = 1800
doc Initial_Landfill_Availability = The value of the landfill availability at the beginning of the simulation period.
unit Initial_Landfill_Availability = items

const Initial_Resources = 90000000
doc Initial_Resources = The value of Natural_Resources at the beginning of the simulation period.
unit Initial_Resources = items

const Inspection_Time = 1
doc Inspection_Time = Time needed for the inspection process.
unit Inspection_Time = week

const Leg_Time = 6*50
doc Leg_Time = Time needed for new environmental regulations.
unit Leg_Time = week

const Minimum_Collection_Percentage = 0.2
doc Minimum_Collection_Percentage = The minimum value of collection percentage.
unit Minimum_Collection_Percentage = (dimensionless)

const Minimum_Limit_of_Recycled_Content = 0.2
doc Minimum_Limit_of_Recycled_Content = The minimum value of recycled content.
unit Minimum_Limit_of_Recycled_Content = (dimensionless)

const Minimum_Recyclability = 0.2
doc Minimum_Recyclability = The minimum value of Recyclability.
unit Minimum_Recyclability = (dimensionless)

const Minimum_Recycling_Percentage = 0.2
doc Minimum_Recycling_Percentage = The minimum value of recycling percentage.
unit Minimum_Recycling_Percentage = (dimensionless)

const Procurement_Time = 2
doc Procurement_Time = Time needed for the supply of Natural_Resources.
unit Procurement_Time = week

const Production_Capacity = 1000
doc Production_Capacity = Capacity of the producer's facilities.
unit Production_Capacity = items/week

const Recyclable_Stock_Keeping_Time = 2
doc Recyclable_Stock_Keeping_Time = The maximum time that Recyclable_Products remain unused in the recycling facilities.
unit Recyclable_Stock_Keeping_Time = week

const Recycling_Capacity = 1000
doc Recycling_Capacity = Capacity of the recycling facilities.
unit Recycling_Capacity = items/week

const Redesign_Time = 12+5*50
doc Redesign_Time = Redesign time.
unit Redesign_Time = week

const RM_Adj_Time = 1
unit RM_Adj_Time = week

const RM_Cover_Time = 2
unit RM_Cover_Time = week

const Shipment_Time = 2
doc Shipment_Time = Time needed to transfer products from producer to distributor.
unit Shipment_Time = week

const SI_Adj_Time = 1
unit SI_Adj_Time = week

```
const   SI_Cover_Time = 2
unit    SI_Cover_Time = week

const   Switch1 = 0
doc     Switch1 = Switch that it can be either 0 or 1.
unit    Switch1 = (dimensionless)

dim     Tactics = (LB=1..4)
const   Tactics = [0, 0, 0, 1]
doc     Tactics = Political tactics introducing new regulatory measures.
unit    Tactics = (dimensionless)
```